PPN calculation for general relativity in xAct

Preliminaries

Load package

This loads the xPPN package into the Mathematica session.

In[1]:=
```
<< xAct`xPPN`
```

```
------------------------------------------------------------
Package xAct`xPerm`  version 1.2.3, {2015, 8, 23}
CopyRight (C) 2003-2020, Jose M. Martin-Garcia, under the General Public License.
Connecting to external linux executable...
Connection established.
------------------------------------------------------------
Package xAct`xTensor`  version 1.1.4, {2020, 2, 16}
CopyRight (C) 2002-2020, Jose M. Martin-Garcia, under the General Public License.
------------------------------------------------------------
These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.
------------------------------------------------------------
** DefMetric: Associating fiducial flat derivative PD to metric.
** DefMetric: Associating fiducial flat derivative PD to metric.
••• OptionValue : Unknown option FlatMetric for DefCovD.
** DefMetric: Associating fiducial flat derivative PD to metric.
   Rules {1} have been declared as UpValues for BkgInvTetradM4.
   Rules {1} have been declared as UpValues for BkgInvTetradS3.
   Rules {1} have been declared as UpValues for BkgTetradM4.
   Rules {1} have been declared as UpValues for BkgInvTetradM4.
   Rules {1} have been declared as UpValues for BkgTetradS3.
   Rules {1} have been declared as UpValues for BkgInvTetradS3.
••• DefMetric : There are already metrics {BkgMetricM4} in vbundle TM4.
```

```
Rules {1} have been declared as UpValues for InvTet .

Rules {1} have been declared as DownValues for TREnergyMomentum .
```

Nicer printing

Use this to automatically convert "dollar" indices in xAct before output is displayed.

```
In[2]:= $PrePrint = ScreenDollarIndices ;
```

Make rules from equations

These functions help to create xAct rules, either with automatic raising / lowering of indices or without.

```
In[3]:= mkrg[eq_Equal] :=
          MakeRule[Evaluate[List @@ eq], MetricOn -> All, ContractMetrics -> True];
        mkr0[eq_Equal] := MakeRule[Evaluate[List @@ eq],
           MetricOn -> None, ContractMetrics -> False];
```

Object definitions

Gravitational constant

The only constant parameter we need to define is the gravitational constant.

```
In[5]:= DefConstantSymbol [kappa, PrintAs -> "κ"];
```

Field Equations

This tensor will hold the gravitational field equations.

```
In[6]:= DefTensor[MetEq[-T4α, -T4β], {MfSpacetime}, Symmetric[{1, 2}], PrintAs -> "ℰ"];
```

Constant coefficients

For convenience, we define some constants, which will be used in the ansatz for solving the field equations.

```
In[7]:= aa[i_] := Module[{sym = Symbol["a" <> ToString[i]]},
          If[! ConstantSymbolQ [sym], DefConstantSymbol [sym,
            PrintAs -> StringJoin["\!\(a\_", ToString[i], "\)"]]]; Return[sym]]
```

Field equations and perturbative expansion

Field equation

Here we define the trace-reversed Einstein equations. These are more convenient to solve, since the time-time component contains only derivatives of the time-time component of the metric, and so the equations decouple immediately.

In[8]:=
```
RicciCD[-T4α, -T4β] - kappa ^2 * TREnergyMomentum [-T4α, -T4β]
```

Out[8]=

$$-\kappa^2 \left(\Theta_{\alpha\beta} - \frac{1}{2}\,\Theta_{\gamma\delta}\ \mathrm{ig}^{\gamma\delta}\ g_{\alpha\beta}\right) + R[\mathring{\nabla}]_{\alpha\beta}$$

Save the equations for later use.

In[9]:=
```
meteqdef = MetEq[-T4α, -T4β] == %;
meteqru = mkr0[meteqdef];
```

3 + 1 split

Split the equations into their time and space components.

In[11]:=
```
{#, # /. meteqru} &[MetEq[-T4α, -T4β]];
ChangeCovD [%, CD, PD];
Expand[%];
SpaceTimeSplits [#, {-T4α -> -T3a, -T4β -> -T3b}] & /@ %;
Expand[%];
Map[ToCanonical , %, {3}];
Map[SortPDs , %, {3}];
meteq31list = %;
meteq31def = Union[Flatten[MapThread[Equal , %, 2]]];
meteq31ru = Flatten[mkrg /@ %];
```

Velocity orders

Expand each component into the respective velocity orders.

In[21]:=
```
Outer[VelocityOrder , meteq31list , Range[0, 4]];
Map[NoScalar , %, {4}];
Expand[%];
Map[ContractMetric [#, OverDerivatives  -> True,
    AllowUpperDerivatives  -> True] &, %, {4}];
Map[ToCanonical , %, {4}];
Map[SortPDs , %, {4}];
meteqvlist  = Simplify[%];
meteqvdef  = Union[Flatten[MapThread[Equal , %, 3]]]
meteqvru  = Flatten[mkrg /@ %];
```

Out[28]=

$$\Big\{\overset{0}{\mathcal{E}}_{00} == 0,\ \overset{1}{\mathcal{E}}_{00} == 0,\ \overset{2}{\mathcal{E}}_{00} == \frac{1}{2}\left(-\kappa^2\rho - \partial_a\partial^a\overset{2}{g}_{00}\right),\ \overset{3}{\mathcal{E}}_{00} == 0,$$

$$\overset{4}{\mathcal{E}}_{00} == \frac{1}{4}\Big(-2\kappa^2\rho\Pi - 6\kappa^2 p - 4\kappa^2\rho\, v_a\, v^a + 4\,\partial_0\partial_a\overset{3}{g}_0{}^a - 2\,\partial_0\partial_0\overset{2}{g}{}^a{}_a - 2\,\partial_a\partial^a\overset{4}{g}_{00} - \partial_a\overset{2}{g}_{00}\,\partial^a\overset{2}{g}_{00} - \partial_a\overset{2}{g}{}^b{}_b\,\partial^a\overset{2}{g}_{00} + 2\,\partial^a\overset{2}{g}_{00}\,\partial_b\overset{2}{g}_a{}^b + 2\kappa^2\rho\,\overset{2}{g}_{00} + 2\,\partial^b\partial^a\overset{2}{g}_{00}\,\overset{2}{g}_{ab}\Big),$$

$$\overset{0}{\mathcal{E}}_{0a} == 0,\ \overset{0}{\mathcal{E}}_{0b} == 0,\ \overset{1}{\mathcal{E}}_{0a} == 0,\ \overset{1}{\mathcal{E}}_{0b} == 0,\ \overset{2}{\mathcal{E}}_{0a} == 0,\ \overset{2}{\mathcal{E}}_{0b} == 0,$$

$$\overset{3}{\mathcal{E}}_{0a} == \frac{1}{2}\left(2\kappa^2\rho\, v_a - \partial_0\partial_a\overset{2}{g}{}^b{}_b + \partial_0\partial_b\overset{2}{g}_a{}^b + \partial_b\partial_a\overset{3}{g}_0{}^b - \partial_b\partial^b\overset{3}{g}_{0a}\right),$$

$$\overset{3}{\mathcal{E}}_{0b} == \frac{1}{2}\left(2\kappa^2\rho\, v_b + \partial_0\partial_a\overset{2}{g}_b{}^a - \partial_0\partial_b\overset{2}{g}{}^a{}_a - \partial_a\partial^a\overset{3}{g}_{0b} + \partial_a\partial_b\overset{3}{g}_0{}^a\right),$$

$$\overset{4}{\mathcal{E}}_{0a} == 0,\ \overset{4}{\mathcal{E}}_{0b} == 0,\ \overset{0}{\mathcal{E}}_{ab} == 0,\ \overset{1}{\mathcal{E}}_{ab} == 0,$$

$$\overset{2}{\mathcal{E}}_{ab} == \frac{1}{2}\left(-\kappa^2\,\delta_{ab}\,\rho + \partial_b\partial_a\overset{2}{g}_{00} - \partial_b\partial_a\overset{2}{g}{}^c{}_c + \partial_c\partial_a\overset{2}{g}_b{}^c + \partial_c\partial_b\overset{2}{g}_a{}^c - \partial_c\partial^c\overset{2}{g}_{ab}\right),$$

$$\overset{3}{\mathcal{E}}_{ab} == 0,\ \overset{4}{\mathcal{E}}_{ab} == \frac{1}{4}\Big(2\kappa^2\,\delta_{ab}\,(-\rho\Pi + p) - 4\kappa^2\rho\, v_a\, v_b - 2\,\partial_0\partial_a\overset{3}{g}_{0b} - 2\,\partial_0\partial_b\overset{3}{g}_{0a} + 2\,\partial_0\partial_0\overset{2}{g}_{ab} + 2\,\partial_b\partial_a\overset{4}{g}_{00} - 2\,\partial_b\partial_a\overset{4}{g}{}^c{}_c + \partial_a\overset{2}{g}_{00}\,\partial_b\overset{2}{g}_{00} + \partial_a\overset{2}{g}{}^{cd}\,\partial_b\overset{2}{g}_{cd} + 2\,\partial_c\partial_a\overset{4}{g}_b{}^c + 2\,\partial_c\partial_b\overset{4}{g}_a{}^c - 2\,\partial_c\partial^c\overset{4}{g}_{ab} + \partial_a\overset{2}{g}_b{}^c\,\partial_c\overset{2}{g}{}^d{}_d + \partial_b\overset{2}{g}_a{}^c\,\partial_c\overset{2}{g}{}^d{}_d - \partial_a\overset{2}{g}_{bc}\,\partial^c\overset{2}{g}_{00} - \partial_b\overset{2}{g}_{ac}\,\partial^c\overset{2}{g}_{00} + \partial_c\overset{2}{g}_{ab}\,\partial^c\overset{2}{g}_{00} - \partial_c\overset{2}{g}{}^d{}_d\,\partial^c\overset{2}{g}_{ab} - 2\,\partial_a\overset{2}{g}_b{}^c\,\partial_d\overset{2}{g}_c{}^d - 2\,\partial_b\overset{2}{g}_a{}^c\,\partial_d\overset{2}{g}_c{}^d + 2\,\partial^c\overset{2}{g}_{ab}\,\partial_d\overset{2}{g}_c{}^d - 2\,\partial_c\overset{2}{g}_{bd}\,\partial^d\overset{2}{g}_a{}^c + 2\,\partial_d\overset{2}{g}_{bc}\,\partial^d\overset{2}{g}_a{}^c + 2\,\partial_b\partial_a\overset{2}{g}_{00}\,\overset{2}{g}_{00} - 2\kappa^2\rho\,\overset{2}{g}_{ab} + 2\,\partial_b\partial_a\overset{2}{g}_{cd}\,\overset{2}{g}{}^{cd} - 2\,\partial_d\partial_a\overset{2}{g}_{bc}\,\overset{2}{g}{}^{cd} - 2\,\partial_d\partial_b\overset{2}{g}_{ac}\,\overset{2}{g}{}^{cd} + 2\,\partial_d\partial_c\overset{2}{g}_{ab}\,\overset{2}{g}{}^{cd}\Big)\Big\}$$

Solution

Check vacuum equations

The zeroth order equations correspond to the vacuum. Check that they are solved for the assumed background.

In[30]:=
```
eqns0  = {PPN[MetEq , 0][-LI[0], -LI[0]], PPN[MetEq , 0][-T3a , -T3b]} /. meteqvru
```

Out[30]=
```
{0, 0}
```

Second order

Extract the second order field equations.

In[31]:=
```
eqns2 =
 FullSimplify[{PPN[MetEq, 2][-LI[0], -LI[0]], PPN[MetEq, 2][-T3a, -T3b]} /. meteqvru]
```

Out[31]= $\left\{\frac{1}{2}\left(-\kappa^2\rho-\partial_a\partial^a \overset{2}{g}_{00}\right), \frac{1}{2}\left(-\kappa^2\ \delta_{ba}\ \rho+\partial_b\partial_a \overset{2}{g}_{00}-\partial_b\partial_a \overset{2}{g}{}^c{}_c+\partial_c\partial_a \overset{2}{g}_b{}^c+\partial_c\partial_b \overset{2}{g}_a{}^c-\partial_c\partial^c \overset{2}{g}_{ab}\right)\right\}$

Define an ansatz for the second order metric perturbations.

In[32]:=
```
ans2def =
 {PPN[Met, 2][-LI[0], -LI[0]] == aa[1] * PotentialU[], PPN[Met, 2][-T3a, -T3b] ==
   aa[2] * PotentialU[] * BkgMetricS3[-T3a, -T3b] + aa[3] * PotentialUU[-T3a, -T3b]}
ans2ru = Flatten[mkrg /@ ans2def];
```

Out[32]= $\left\{\overset{2}{g}_{00} == a_1\ U,\ \overset{2}{g}_{ab} == a_2\ \delta_{ab}\ U + a_3\ U_{ab}\right\}$

Insert the ansatz into the field equations and convert derivatives of the PPN potentials into matter source terms.

In[34]:=
```
eqns2 /. ans2ru;
PotentialUToChi /@ %;
PotentialUUToChi /@ %;
Expand[%];
ToCanonical /@ %;
ContractMetric[#, OverDerivatives -> True, AllowUpperDerivatives -> True] & /@
  %;
PotentialToSource /@ %;
Expand[%];
ToCanonical /@ %;
SortPDs /@ %;
eqnsa2 = FullSimplify[%]
```

Out[44]= $\left\{-\frac{1}{2}\left(\kappa^2-4\ a_1\ \pi\right)\rho,\ \frac{1}{4}\left(-2\left(\kappa^2-4\left(a_2+a_3\right)\pi\right)\ \delta_{ab}\ \rho+\left(-a_1+a_2+a_3\right)\partial_c\partial^c\partial_b\partial_a\chi\right)\right\}$

The equations to be solved are extracted as coefficients of the matter terms. The last condition is the gauge condition, which mandates that the spatial part of the metric should be diagonal.

In[45]:=
```
eqnsc2 = FullSimplify[{Coefficient[eqnsa2[[1]], Density[]],
   Coefficient[eqnsa2[[2]], Density[] * BkgMetricS3[-T3a, -T3b]], aa[3]}]
```

Out[45]= $\left\{-\frac{\kappa^2}{2}+2\ a_1\ \pi,\ -\frac{\kappa^2}{2}+2\left(a_2+a_3\right)\pi,\ a_3\right\}$

Solve for the constant coefficients in the equations.

In[46]:=
```
sola2 = FullSimplify[First[Solve[# == 0 & /@ eqnsc2, aa /@ Range[1, 3]]]]
```

Out[46]= $\left\{a_1\to\frac{\kappa^2}{4\ \pi},\ a_2\to\frac{\kappa^2}{4\ \pi},\ a_3\to 0\right\}$

Check that the solution indeed solves the component equations.

In[47]:=
```
Simplify[eqnsa2 /. sola2]
```

Out[47]= {0, 0}

Insert the solution for the constant coefficients into the ansatz, to obtain the solution for the metric components.

In[48]:=
```
sol2def = ans2def /. sola2
sol2ru = Flatten[mkrg /@ sol2def];
```

Out[48]= $\left\{ \overset{2}{g}_{00} == \frac{\kappa^2 U}{4\pi}, \overset{2}{g}_{ab} == \frac{\kappa^2 \delta_{ab} U}{4\pi} \right\}$

Check that the metric components we found indeed solve the second order field equations.

In[50]:=
```
eqns2 /. sol2ru;
Expand[%];
PotentialToSource /@ %;
ToCanonical /@ %;
SortPDs /@ %;
Simplify[%]
```

Out[55]= {0, 0}

Third order

Extract the third order field equations.

In[56]:=
```
eqns3 = FullSimplify[PPN[MetEq, 3][-LI[0], -T3a] /. meteqvru]
```

Out[56]= $\frac{1}{2}\left(2\kappa^2 \rho\, v_a - \partial_0\partial_a \overset{2}{g}{}^b{}_b + \partial_0\partial_b \overset{2}{g}{}_a{}^b + \partial_b\partial_a \overset{3}{g}{}_0{}^b - \partial_b\partial^b \overset{3}{g}{}_{0a}\right)$

Define an ansatz for the third order metric perturbations.

In[57]:=
```
ans3def =
  PPN[Met, 3][-LI[0], -T3a] == aa[4] * PotentialV[-T3a] + aa[5] * PotentialW[-T3a]
ans3ru = mkrg[ans3def];
```

Out[57]= $\overset{3}{g}_{0a} == a_4 V_a + a_5 W_a$

Insert the ansatz into the field equations and convert derivatives of the PPN potentials into matter source terms.

In[59]:=
```
eqns3 /. ans3ru /. sol2ru;
PotentialWToChiV[%];
Expand[%];
ContractMetric[%, OverDerivatives -> True, AllowUpperDerivatives -> True];
PotentialChiToU[%];
PotentialVToU[%];
PotentialToSource[%];
ToCanonical[%];
SortPDs[%];
eqnsa3 = FullSimplify[%]
```

Out[68]=

$$\frac{\left(\kappa^2 + 2\,(a_4 + a_5)\,\pi\right)\left(4\,\pi\,\rho\;v_a - \partial_0\partial_a U\right)}{4\,\pi}$$

Solve for the constant coefficients in the equations, keeping a gauge freedom, which is left up to the fourth velocity order.

In[69]:=
```
sola3 = FullSimplify[First[Solve[{eqnsa3 == 0, aa[5] - aa[4] == aa[0]}, {aa[5], aa[4]}]]]
```

Out[69]=

$$\left\{a_5 \to -\frac{\kappa^2 - 2\,a_0\,\pi}{4\,\pi},\; a_4 \to -\frac{\kappa^2 + 2\,a_0\,\pi}{4\,\pi}\right\}$$

Check that the solution indeed solves the component equations.

In[70]:=
```
Simplify[eqnsa3 /. sola3]
```

Out[70]= 0

Insert the solution for the constant coefficients into the ansatz, to obtain the solution for the metric components.

In[71]:=
```
sol3def = ans3def /. sola3
sol3ru = mkrg[sol3def];
```

Out[71]=

$$\overset{3}{g}_{0a} == -\frac{\left(\kappa^2 + 2\,a_0\,\pi\right)\,V_a}{4\,\pi} - \frac{\left(\kappa^2 - 2\,a_0\,\pi\right)\,W_a}{4\,\pi}$$

Check that the metric components we found indeed solve the third order field equations.

In[73]:=
```
eqns3 /. sol2ru /. sol3ru;
PotentialWToChiV[%];
Expand[%];
ContractMetric[%, OverDerivatives -> True, AllowUpperDerivatives -> True];
PotentialChiToU[%];
PotentialVToU[%];
PotentialToSource[%];
ToCanonical[%];
SortPDs[%];
Simplify[%]
```

Out[82]= 0

Fourth order

Extract the fourth order field equations.

In[83]:=
```
eqns4 = PPN[MetEq, 4][-LI[0], -LI[0]] /. meteqvru
```

Out[83]=

$$-\frac{1}{2}\kappa^2\rho\Pi-\frac{3\kappa^2 p}{2}-\kappa^2\rho\, v_a\; v^a+\partial_0\partial_a\overset{3}{g}{}_0{}^a-\frac{1}{2}\partial_0\partial_0\overset{2}{g}{}^a{}_a-\frac{1}{2}\partial_a\partial^a\overset{4}{g}{}_{00}-\frac{1}{4}\partial_a\overset{2}{g}{}_{00}\,\partial^a\overset{2}{g}{}_{00}-\frac{1}{4}\partial_a\overset{2}{g}{}^b{}_b\,\partial^a\overset{2}{g}{}_{00}+\frac{1}{2}\partial^a\overset{2}{g}{}_{00}\,\partial_b\overset{2}{g}{}_a{}^b+\frac{1}{2}\kappa^2\rho\,\overset{2}{g}{}_{00}+\frac{1}{2}\partial^b\partial^a\overset{2}{g}{}_{00}\,\overset{2}{g}{}_{ab}$$

Define an ansatz for the fourth order metric perturbations.

In[84]:=
```
ans4def =
 PPN[Met, 4][-LI[0], -LI[0]] == aa[6] * PotentialPhi1[] + aa[7] * PotentialPhi2[] +
   aa[8] * PotentialPhi3[] + aa[9] * PotentialPhi4[] + aa[10] * PotentialU[]^2
ans4ru = mkrg[ans4def];
```

Out[84]=

$$\overset{4}{g}{}_{00} == a_6\,\Phi_1+a_7\,\Phi_2+a_8\,\Phi_3+a_9\,\Phi_4+a_{10}\,U^2$$

Insert the ansatz into the field equations and convert derivatives of the PPN potentials into matter source terms.

In[86]:=
```
eqns4 /. ans4ru /. sol2ru /. sol3ru;
Expand[%];
ContractMetric[%, OverDerivatives -> True, AllowUpperDerivatives -> True];
PotentialVToU[%];
PotentialWToU[%];
PotentialToSource[%];
ToCanonical[%];
SortPDs[%];
Expand[%];
eqnsa4 = Simplify[ScreenDollarIndices[%]]
```

Out[95]=

$$-\frac{1}{2}\kappa^2\rho\Pi+2a_8\pi\rho\Pi+4a_{10}\pi\rho U+2a_7\pi\rho U-\frac{3\kappa^2 p}{2}+2a_9\pi p-\left(\kappa^2-2a_6\pi\right)\rho\, v_a\; v^a+\left(a_0-\frac{3\kappa^2}{8\pi}\right)\partial_0\partial_0 U-a_{10}\,\partial_a U\,\partial^a U-\frac{\kappa^4\,\partial_a U\,\partial^a U}{32\pi^2}$$

Coefficient of the pressure.

In[96]:=
```
eq1 = Simplify[Coefficient[eqnsa4, Pressure[]]]
```

Out[96]=

$$-\frac{3\kappa^2}{2}+2a_9\pi$$

Coefficient of the internal energy.

In[97]:=
```
eq2 = Simplify[Coefficient[eqnsa4, Density[] * InternalEnergy[]]]
```

Out[97]=

$$-\frac{\kappa^2}{2}+2a_8\pi$$

Coefficient of the gravitational potential energy.

In[98]:=
```
eq3 = Simplify[Coefficient[eqnsa4, Density[] * PotentialU[]]]
```

Out[98]= $2\,(2\,a_{10} + a_7)\,\pi$

Coefficient of the second time derivative.

In[99]:=
```
eq4 = Simplify[Coefficient[eqnsa4, ParamD[TimePar, TimePar][PotentialU[]]]]
```

Out[99]= $a_0 - \frac{3\,\kappa^2}{8\,\pi}$

Coefficient of the kinetic energy.

In[100]:=
```
eq5 = Simplify[Coefficient[eqnsa4, Density[] * Velocity[-T3a] * Velocity[T3a]]]
```

Out[100]= $-\kappa^2 + 2\,a_6\,\pi$

Coefficient of the potential term.

In[101]:=
```
eq6 = Simplify[Coefficient[eqnsa4, PD[-T3a][PotentialU[]] * PD[T3a][PotentialU[]]]]
```

Out[101]= $-a_{10} - \frac{\kappa^4}{32\,\pi^2}$

Check that we have fully decomposed the equations.

In[102]:=
```
Simplify[Pressure[] * eq1 + Density[] * InternalEnergy[] * eq2 +
  Density[] * PotentialU[] * eq3 + ParamD[TimePar, TimePar][PotentialU[]] * eq4 +
  Density[] * Velocity[-T3a] * Velocity[T3a] * eq5 +
  PD[-T3a][PotentialU[]] * PD[T3a][PotentialU[]] * eq6 - eqnsa4]
```

Out[102]= 0

Solve for the constant coefficients in the equations.

In[103]:=
```
sola4 = Simplify[First[Solve[
    # == 0 & /@ {eq1, eq2, eq3, eq4, eq5, eq6}, aa /@ Prepend[Range[6, 10], 0]]]]
```

Out[103]= $\left\{a_0 \to \frac{3\,\kappa^2}{8\,\pi},\ a_6 \to \frac{\kappa^2}{2\,\pi},\ a_7 \to \frac{\kappa^4}{16\,\pi^2},\ a_8 \to \frac{\kappa^2}{4\,\pi},\ a_9 \to \frac{3\,\kappa^2}{4\,\pi},\ a_{10} \to -\frac{\kappa^4}{32\,\pi^2}\right\}$

Check that the solution indeed solves the component equations.

In[104]:=
```
Simplify[eqnsa4 /. sola4]
```

Out[104]= 0

Enhance the third order solution, using the gauge fixing condition determined at the fourth order.

In[105]:=
```
sol3def = ans3def /. Simplify[sola3 /. sola4]
sol3ru = mkrg[sol3def];
```

Out[105]= $\overset{3}{g}_{0a} == -\frac{7\,\kappa^2\,V_a}{16\,\pi} - \frac{\kappa^2\,W_a}{16\,\pi}$

Insert the solution for the constant coefficients into the ansatz, to obtain the solution for the metric components.

In[107]:=
```
sol4def = ans4def /. sola4
sol4ru = mkrg[sol4def];
```

Out[107]=
$$\overset{4}{g}_{00} == \frac{\kappa^2 \Phi_1}{2\pi} + \frac{\kappa^4 \Phi_2}{16\pi^2} + \frac{\kappa^2 \Phi_3}{4\pi} + \frac{3\kappa^2 \Phi_4}{4\pi} - \frac{\kappa^4 U^2}{32\pi^2}$$

Check that the metric components we found indeed solve the fourth order field equations.

In[109]:=
```
eqns4 /. sol2ru /. sol3ru /. sol4ru;
Expand[%];
ContractMetric[%, OverDerivatives -> True, AllowUpperDerivatives -> True];
PotentialVToU[%];
PotentialWToU[%];
PotentialToSource[%];
ToCanonical[%];
SortPDs[%];
Expand[%];
Simplify[%]
```

Out[118]= 0

PPN metric and parameters

PPN metric

To read off the PPN parameters, we use the following metric components.

In[119]:=
```
metcomp = {PPN[Met, 2][-LI[0], -LI[0]], PPN[Met, 2][-T3a, -T3b],
  PPN[Met, 3][-LI[0], -T3a], PPN[Met, 4][-LI[0], -LI[0]]}
```

Out[119]=
$$\{\overset{2}{g}_{00}, \overset{2}{g}_{ab}, \overset{3}{g}_{0a}, \overset{4}{g}_{00}\}$$

Insert the solution we obtained into the metric components.

In[120]:=
```
metcomp /. sol2ru /. sol3ru /. sol4ru;
ToCanonical[%];
Expand[%];
ppnmet = Simplify[%];
metdef = MapThread[Equal, {metcomp, %}, 1]
```

Out[124]=
$$\left\{\overset{2}{g}_{00} == \frac{\kappa^2 U}{4\pi}, \overset{2}{g}_{ab} == \frac{\kappa^2 \delta_{ab} U}{4\pi}, \overset{3}{g}_{0a} == -\frac{\kappa^2 (7 V_a + W_a)}{16\pi}, \overset{4}{g}_{00} == \frac{8\kappa^2 \pi (2\Phi_1 + \Phi_3 + 3\Phi_4) + \kappa^4 (2\Phi_2 - U^2)}{32\pi^2}\right\}$$

For later comparison, convert the selected components also to their standard form in terms of PPN parameters and potentials.

In[125]:=
```
stamet = Simplify[MetricToStandard /@ metcomp]
```

Out[125]=
$$\left\{2\,U,\ 2\,\gamma\,\delta_{ba}\,U,\ \frac{1}{2}\left(-(3+\alpha_1-\alpha_2+4\,\gamma-2\,\xi+\zeta_1)\,V_a-(1+\alpha_2+2\,\xi-\zeta_1)\,W_a\right),\ 2\,\Phi_1+\alpha_3\,\Phi_1+2\,\gamma\,\Phi_1+\zeta_1\,(-\mathcal{A}+\Phi_1)+2\,\Phi_2-4\,\beta\,\Phi_2+6\,\gamma\,\Phi_2+2\,\zeta_2\,\Phi_2+2\,\Phi_3+2\,\zeta_3\,\Phi_3+6\,\gamma\,\Phi_4+6\,\zeta_4\,\Phi_4+2\,\xi\,(\mathcal{A}-\Phi_1+\Phi_2-2\,\Phi_4-\Phi_W)-2\,\beta\,U^2\right\}$$

Newtonian gravitational constant

To solve for the gravitational constant, compare the second order result with the standard normalization.

In[126]:=
```
kappaeq = First[ppnmet] == First[stamet]
```

Out[126]=
$$\frac{\kappa^2\,U}{4\,\pi} == 2\,U$$

When solving for κ, make sure to catch the positive root.

In[127]:=
```
kappadef =
  kappa == First[Sqrt[FullSimplify[k2 /. Solve[kappaeq /. kappa -> Sqrt[k2], k2]]]]
kapparu = mkrg[kappadef];
```

Out[127]=
$$\kappa == 2\,\sqrt{2\,\pi}$$

PPN parameters

To generate equations for the PPN parameters, compare the obtained solution with the standard PPN metric.

In[129]:=
```
pareqns = Simplify[ToCanonical[stamet - ppnmet /. kapparu]]
```

Out[129]=
$$\left\{0,\ 2\,(-1+\gamma)\,\delta_{ab}\,U,\ \frac{1}{2}\left(-(-4+\alpha_1-\alpha_2+4\,\gamma-2\,\xi+\zeta_1)\,V_a+(-\alpha_2-2\,\xi+\zeta_1)\,W_a\right),\ -2\,\Phi_1+\alpha_3\,\Phi_1+2\,\gamma\,\Phi_1+\zeta_1\,(-\mathcal{A}+\Phi_1)-2\,\Phi_2-4\,\beta\,\Phi_2+6\,\gamma\,\Phi_2+2\,\zeta_2\,\Phi_2+2\,\zeta_3\,\Phi_3-6\,\Phi_4+6\,\gamma\,\Phi_4+6\,\zeta_4\,\Phi_4+2\,\xi\,(\mathcal{A}-\Phi_1+\Phi_2-2\,\Phi_4-\Phi_W)+2\,U^2-2\,\beta\,U^2\right\}$$

We will consider the coefficients in front of the following potentials.

In[130]:=
```
pots = {PotentialU[] * BkgMetricS3[-T3a, -T3b], PotentialV[-T3a],
   PotentialW[-T3a], PotentialA[], PotentialU[]^2, PotentialPhiW[],
   PotentialPhi1[], PotentialPhi2[], PotentialPhi3[], PotentialPhi4[]}
```

Out[130]=
$$\{\delta_{ab}\,U,\ V_a,\ W_a,\ \mathcal{A},\ U^2,\ \Phi_W,\ \Phi_1,\ \Phi_2,\ \Phi_3,\ \Phi_4\}$$

Extract the coefficients from the difference between our result and the standard PPN metric. These terms must vanish.

In[131]:=
```
eqns = DeleteCases[Flatten[Simplify[Outer[Coefficient, pareqns, pots]]], 0]
```

Out[131]=

$$\left\{2(-1+\gamma),\ \frac{1}{2}(4-\alpha_1+\alpha_2-4\gamma+2\xi-\zeta_1),\ \frac{1}{2}(-\alpha_2-2\xi+\zeta_1),\ 2\xi-\zeta_1,\ 2-2\beta,\ -2\xi,\ -2+\alpha_3+2\gamma-2\xi+\zeta_1,\ 2(-1-2\beta+3\gamma+\xi+\zeta_2),\ 2\zeta_3,\ -6+6\gamma-4\xi+6\zeta_4\right\}$$

List of PPN parameters we are solving for.

In[132]:=
```
pars = {ParameterBeta, ParameterGamma, ParameterXi,
  ParameterAlpha1, ParameterAlpha2, ParameterAlpha3,
  ParameterZeta1, ParameterZeta2, ParameterZeta3, ParameterZeta4}
```

Out[132]=

$$\{\beta, \gamma, \xi, \alpha_1, \alpha_2, \alpha_3, \zeta_1, \zeta_2, \zeta_3, \zeta_4\}$$

Finally, solve the equations and determine the PPN parameters.

In[133]:=
```
parsol = FullSimplify[Solve[# == 0 & /@ eqns, pars][[1]]]
```

Out[133]=

$$\{\beta \to 1, \gamma \to 1, \xi \to 0, \alpha_1 \to 0, \alpha_2 \to 0, \alpha_3 \to 0, \zeta_1 \to 0, \zeta_2 \to 0, \zeta_3 \to 0, \zeta_4 \to 0\}$$